

Coordinate geografiche e cartesiane geocentriche: loro trasformazione
di Michele T. Mazzucato

Si tratta di una trasformazione di coordinate, sia diretta sia inversa, molto importante in geodesia. Il passaggio diretto, da coordinate geografiche a quelle cartesiane ortogonali nello spazio, non presenta particolare difficoltà ed è effettuabile mediante le note formule qui di seguito riportate:

**Trasformazione di coordinate
da geografiche ($\varphi;\lambda;h$) a cartesiane ortogonali nello spazio ($X;Y;Z$)
problema diretto**

$$X = (N + h) \times \cos\varphi \times \cos\lambda$$

$$Y = (N + h) \times \cos\varphi \times \sin\lambda$$

$$Z = (N - e^2 \times N + h) \times \sin\varphi$$

$$\text{con } e^2 = \frac{a^2 - b^2}{a^2} \quad \text{e} \quad N = \frac{a}{\sqrt{(1 - e^2 \times \sin^2\varphi)}}$$

dove

a = raggio equatoriale

b = raggio polare

e = eccentricità prima

N = raggio di curvatura della sezione in primo verticale (gran normale)

Così non si può certo dire per il passaggio inverso, ossia da cartesiane ortogonali nello spazio a geografiche, che presenta alcune difficoltà dovute principalmente al fatto che N dipende dalla latitudine. Per quest'ultimo caso numerose sono le soluzioni proposte in letteratura sia di tipo iterativo come quella proposta da W.A. HEISKANEN - H. MORITZ (1967) oppure ricorrendo alla risoluzione di equazioni algebriche di 4° grado tramite procedure di calcolo numeriche come quelle di V. TOMELLIERI (1970), di M. UNGUENDOLI (1974) e di E. KRAKIWSKY - P. VANIČEK (1982) sia di tipo chiuso come quella di P. BENCINI (1968), di M.K. PAUL (1973), di J.Y. CHEN (1981) e di K.M. BORKOWSKI (1989), solo per citarne qualcuna. Tra le tante soluzioni si riporta il semplice metodo non iterativo proposto da Bowring nel 1976, per la latitudine, e nel 1985, per l'altezza:

**Trasformazione di coordinate
da cartesiane ortogonali nello spazio ($X;Y;Z$) a geografiche ($\varphi;\lambda;h$)
problema inverso**

Formule di B.R. BOWRING, 1976 per il calcolo della latitudine φ

$$\text{tg}\varphi = \frac{(Z + e'^2 \times b \times \sin^3\mu)}{(r - e^2 \times a \times \cos^3\mu)} \quad [1]$$

$$\text{con } e'^2 = \frac{a^2 - b^2}{b^2}, \quad \text{tg}\mu = \frac{a \times Z}{b \times r} \quad \text{e} \quad r = \sqrt{X^2 + Y^2}$$

dove

e' = eccentricità seconda

r = raggio del parallelo

μ = latitudine ridotta

Formule di B.R. BOWRING, 1985 per il calcolo dell'altezza h

$$h = r \times \cos\varphi + Z \times \sin\varphi - a\sqrt{(1 - e^2 \times \sin^2\varphi)} \quad [2]$$

Per il calcolo della longitudine λ si utilizza la seguente espressione che resta valida per tutte le soluzioni proposte

$$\operatorname{tg}\lambda = \frac{Y}{X}$$

Alle nostre latitudini essa fornisce direttamente il valore della longitudine a est da Greenwich mentre per il caso generale occorre, ovviamente, tener conto dei segni algebrici delle coordinate cartesiane ortogonali nello spazio X e Y ricavabili dalla sottostante tabella nella quale viene dato l'intervallo di variabilità della longitudine:

Y	+	+	-	-
X	+	-	+	-
long	est 0°-90°	est 90°-180°	ovest 0°-90°	ovest 90°-180°

Mentre con la seguente formula, proposta da E. KRAKIVSKY - P. VANIČEK (1982), si può calcolare la longitudine senza bisogno di definirne il segno e quindi il suo fuso:

$$\lambda = 2\arctg\left(\frac{Y}{X + r}\right) \quad [3]$$

Naturalmente si possono realizzare semplici programmi informatici come quello del quale viene fornito il seguente listato che utilizza, in particolare, le formule [1], [2] e [3] per la trasformazione inversa.

listato del programma in BASIC

```

10 REM Realizzazione di M.T. Mazzucato 2001 [CARTESIA.bas]
20 REM Programma per la trasformazione diretta ed inversa tra le
30 REM coordinate geografiche e quelle cartesiane
40 CLEAR: DEFDBL A-Z: KEY OFF: COLOR 15,1: CLS: PRINT
50 REM parametri ellissoide
60 A= 6378388!: B= 6356911.946#: E1= .006722670022#: E2= .006768170197#
70 REM valore di pi greco
80 P= 3.141592653589793#
90 PRINT "Trasformazione di coordinate"
100 PRINT: PRINT "<a> geografiche => cartesiane (diretta)"
110 PRINT: PRINT "<b> cartesiane => geografiche (inversa)"
120 PRINT: PRINT: PRINT "<c> uscita"
130 A$=INKEY$
140 IF A$="a" THEN GOTO 180
150 IF A$="b" THEN GOTO 350
160 IF A$="c" THEN CLS: IF A$="c" THEN SYSTEM
170 IF A$<>"a" THEN GOTO 130
180 CLS: PRINT
190 PRINT "geografiche => cartesiane": PRINT
200 PRINT: INPUT "- latitudine (x.xx°)=";LAT
210 PRINT: INPUT "- longitudine (x.xx°)=";LONGITUDINE
220 PRINT: INPUT "- altitudine (m)=";H
230 PRINT: PRINT: INPUT "Vuoi cambiare (s/n)=";A$
240 IF A$="s" THEN GOTO 180
250 IF A$="n" THEN GOTO 270
260 IF A$<>"s" THEN GOTO 230

```

```

270 IF LAT=0 AND LONGITUDINE=0 AND H=0 THEN GOTO 550
280 REM esecuzione dei calcoli
290 LATR=LAT*P/180: LONGITUDINER=LONGITUDINE*P/180
300 N= A/(SQR (1-E1*(SIN (LATR))^2))
310 X= (N+H)*COS (LATR)*COS (LONGITUDINER)
320 Y= (N+H)*COS (LATR)*SIN (LONGITUDINER)
330 Z= (N-E1*N+H)*SIN (LATR)
340 GOTO 550
350 CLS: PRINT
360 PRINT "cartesiane => geografiche": PRINT
370 PRINT: INPUT "- coordinata X (m)=";X
380 PRINT: INPUT "- coordinata Y (m)=";Y
390 PRINT: INPUT "- coordinata Z (m)=";Z
400 PRINT: PRINT: INPUT "Vuoi cambiare (s/n)=";A$
410 IF A$="s" THEN GOTO 350
420 IF A$="n" THEN GOTO 440
430 IF A$<>"s" THEN GOTO 400
440 IF X=0 AND Y=0 AND Z=0 THEN GOTO 550
450 REM esecuzione dei calcoli
460 REM Formula di B.R. Bowring, 1976 per la latitudine
470 R= SQR (X^2+Y^2)
480 MICRON= ATN ((A*Z)/(B*R))
490 LATR= ATN ((Z+E2*B*(SIN (MICRON))^3) / (R-E1*A*(COS (MICRON))^3))
500 REM Formula di E. Krakiwsky - P. Vanicek, 1982 per la longitudine
510 LONGITUDINER= 2*ATN (Y/(X+R))
520 REM Formula di B.R. Bowring, 1985 per h
530 H= R*COS (LATR)+Z*SIN (LATR)-A*SQR (1-E1*(SIN (LATR))^2)
540 LAT=LATR*180/P: LONGITUDINE= LONGITUDINER*180/P
550 CLS: PRINT: PRINT "Risultati:": PRINT
560 PRINT "latitudine   =";USING "###.#####°";LAT
570 PRINT "longitudine  =";USING "###.#####°";LONGITUDINE
580 PRINT "altitudine   =";USING "#####.### m";H: PRINT
590 PRINT "coordinata X =";USING "#####.### m";X
600 PRINT "coordinata Y =";USING "#####.### m";Y
610 PRINT "coordinata Z =";USING "#####.### m";Z
620 PRINT: PRINT: PRINT "<n>= nuova trasformazione          <u>= uscita"
630 A$=INKEY$
640 IF A$="n" THEN GOTO 10
650 IF A$="u" THEN COLOR 7,0: CLS: IF A$="u" THEN SYSTEM
660 IF A$<>"n" THEN GOTO 630

```

nota: è possibile utilizzare un qualsiasi ellissoide di riferimento sostituendo semplicemente i parametri ellissoidici presenti nella RIGA 60 relativi all'Ellissoide Internazionale (Hayford, 1924).

Bibliografia

- BARBARELLA M.; GATTI M., *Note sulla trasformazione da sistema geocentrico a sistema ellissoidico*, Bollettino di Geodesia e Scienze Affini n.2/1993 pp.109-132 IGM, Firenze
- BOWRING B.R., *Transformation from spatial to geographical coordinates*, Survey Review vol. 23 July 1976 pp. 323-327
- BOWRING B.R., *The accuracy of geodetic latitude and height equations*, Survey Review vol. 28 October 1985
- MAZZUCATO M.T., *Globo terrestre*, BIROMA Galliera V. Padova (1996)
- MAZZUCATO M.T., *La Figura della Terra*, CLUP Milano (2003)
- STRANG VAN HEES G.L., *Method Bowring for the computation of latitude ϕ from rectangular coordinates (X,Y,Z)*, Bulletin n. 5/dic 1996 pp. 25-30 International Geoid Service, Milano